IMS Health & Quintiles are now

IQVIA®

Contrast Statements in statistical modelling

28th March 2019

Marta Bacmaga

+ Contrast Statements
+ Examples
+ SAS code
+ Checking contrast statement
+ Results interpretation

Contrast Statements

- Help to test univariate hypothesis, i.e.
  - population mean for treatment 1 is the same as population mean for treatment 2

$$\mathbf{H_o}: \boldsymbol{\mu_1} = \boldsymbol{\mu_2}$$

  - average of population mean for treatments 1,2 and 3 is the same as population mean for treatment 4

$$\mathbf{H_o}: \frac{\boldsymbol{\mu_1} + \boldsymbol{\mu_2} + \boldsymbol{\mu_3}}{\mathbf{3}} = \boldsymbol{\mu_4}$$

- Can be used in many different analysis, i.e. ANCOVA and MMRM models
- They need to add up to 0.
- Normally work with 1/-1/0, unless we have group variables.

Examples: ANCOVA

Assume a linear model for 4 treatments:

$Y = A + B_1TRT1 + B_2TRT2 + B_3TRT3 + B_4TRT4$

Comparisons:

TRT1 vs. TRT2 : $B_1 - B_2$ -> 1 -1 0 0

TRT1 vs. TRT3 : $B_1 - B_3$ -> 1 0 -1 0

~~**TRT1+TRT2+TRT3 vs TRT4:** $B_1 + B_2 + B_3 - B_4$ -> 1 1 1 -1 ->NOT EQUAL TO ZERO!~~

TRT1+TRT2+TRT3 vs TRT4: $(B_1 - B_4) + (B_2 - B_4) + (B_3 - B_4) = B_1 + B_2 + B_3 - 3B_4$ -> 1 1 1 -3

SAS Code

ANCOVA

```
proc mixed data=data1;
   class trt tansgr1n racegr1n wgtgrbln;
   model chg=trt tansgr1n racegr1n wgtgrbln base;
   LSMESTIMATE trt “TRT1 vs TRT4” 1 0 0 -1 /cl e;
run;
```

```
proc mixed data=data1;
   class trt tansgr1n racegr1n wgtgrbln;
   model chg=trt tansgr1n racegr1n wgtgrbln base;
   ESTIMATE “TRT1 vs TRT4” trt 1 0 0 -1 /cl e;
run;
```

THE SAME RESULT

```
proc mixed data=data1;
   class trt tansgr1n racegr1n wgtgrbln;
   model base=trt tansgr1n racegr1n wgtgrbln base;
   CONTRAST “TRT1 vs TRT4” trt 1 0 0 -1 / e;
run;
```

ONLY P-VALUE

Validity

	LMATRIX	EFFECT	TRT	TANSGR1N	RACEGR1N	WGTGRBLN	ROW1
1	1	Intercept	_	_	_	_	0
2	1	TRT	1	_	_	_	1
3	1	TRT	2	_	_	_	0
4	1	TRT	3	_	_	_	0
5	1	TRT	4	_	_	_	-1
6	1	TANSGR1N	_	1	_	_	0
7	1	TANSGR1N	_	2	_	_	0
8	1	RACEGR1N	_	_	1	_	0
9	1	RACEGR1N	_	_	2	_	0
10	1	WGTGRBLN	_	_	_	1	0
11	1	WGTGRBLN	_	_	_	2	0
12	1	BASE	_	_	_	_	0

Interpretation of results

$\mathbf{H_0}: \boldsymbol{\mu_1} = \boldsymbol{\mu_4}$
$\mathbf{H_1}: \boldsymbol{\mu_1} \neq \boldsymbol{\mu_4}$

Significance level: 5%
P-value: 0.5983>0.05
=> Not enough evidence to Reject H_0.

LSMESTIMATE

	STMTNO	EFFECT	LABEL	ESTIMATE	STDERR	DF	TVALUE	PROBT	ALPHA	LOWER	UPPER
1	1	TRT	TRT1 vs TRT4	-1.3584	2.5668	71	-0.53	0.5983	0.05	-6.4764	3.7597

ESTIMATE

	LABEL	ESTIMATE	STDERR	DF	TVALUE	PROBT	ALPHA	LOWER	UPPER
1	TRT1 vs TRT4	-1.3584	2.5668	71	-0.53	0.5983	0.05	-6.4764	3.7597

CONTRAST

	LABEL	NUMDF	DENDF	FVALUE	PROBF
1	TRT1 vs TRT4	1	71	0.28	0.5983

Examples: MMRM

Assume a linear model for 4 treatments and 3 visits:

$$Y = A + B_1TRT1 + B_2TRT2 + B_3TRT3 + B_4TRT4 + C_1VIS1 + C_2VIS2 + C_3VIS3$$
$$+ D_1TRT1VIS1 + D_2TRT1VIS2 + D_3TRT1VIS3 + D_4TRT2VIS1 + D_5TRT2VIS2 + D_6TRT2VIS3 +$$
$$D_7TRT3VIS1 + D_8TRT3VIS2 + D_9TRT3VIS3 + D_{10}TRT4VIS1 + D_{11}TRT4VIS2 + D_{12}TRT4VIS3$$

Comparisons:

TRT1 vs. TRT2 at Visit 1 : $B_1 + C_1 + D_1 - B_2 - C_1 - D_4$

TRT: $B_1 - B_2$ -> 1 -1 0 0

TRT*VIS: D_1-D_4 -> 1 0 0 -1 0 0 0 0 0 0 0 0

TRT1+TRT2+TRT3 vs TRT4 at Visit 1 :

$(B_1 + C_1 + D_1 - B_4 - C_1 - D_{10}) + (B_2 + C_1 + D_4 - B_4 - C_1 - D_{10}) + (B_3 + C_1 + D_7 - B_4 - C_1 - D_{10}) =$
$B_1 + D_1 + B_2 + D_4 + B_3 + D_7 - 3B_4 - 3D_{10}$

TRT: $B_1 + B_2 + B_3 - 3B_4$ -> 1 1 1 -3

TRT*VIS: $D_1 + D_4 + D_7 - 3D_{10}$ -> 1 0 0 1 0 0 1 0 0 -3 0 0

SAS Code

MMRM

```
proc mixed data=data1;
   class trt vis tansgr1n racegr1n wgtgrbln;
   model y=trt  vis trt*vis tansgr1n racegr1n wgtgrbln;
   repeated vis / type=un subject=subject;
   LSMESTIMATE trt*vis “TRT1+TRT2+TRT3 vs TRT4 at VIS1” 1 0 0 1 0 0 1 0 0 -3 0 0 /cl e divisor=3;
run;

proc mixed data=data1;
   class trt vis tansgr1n racegr1n wgtgrbln;
   model y=trt vis trt*vis tansgr1n racegr1n wgtgrbln;
   repeated vis / type=un subject=subject;
   ESTIMATE “TRT1+TRT2+TRT3 vs TRT4 at VIS1” trt 1 1 1 -3
                                    trt*vis 1 0 0 1 0 0 1 0 0 -3 0 0 /cl e divisor=3;
run;

proc mixed data=data1;
   class trt vis tansgr1n racegr1n wgtgrbln;
   model y=trt vis trt*vis tansgr1n racegr1n wgtgrbln;
   repeated vis / type=un subject=subject;
   CONTRAST “TRT1+TRT2+TRT3 vs TRT4 at VIS1” trt 1 1 1 -3
                                    trt*vis 1 0 0 1 0 0 1 0 0 -3 0 0 /e;
run;
```



Validity

	LMATRIX	EFFECT	TRT	VIS	TANSGR1N	RACEGR1N	WGTGRBLN	ROW1
1	1	Intercept	_	_	_	_	_	0
2	1	TRT	1	_	_	_	_	0.3333
3	1	TRT	2	_	_	_	_	0.3333
4	1	TRT	3	_	_	_	_	0.3333
5	1	TRT	4	_	_	_	_	-1
6	1	VIS	_	5	_	_	_	0
7	1	VIS	_	6	_	_	_	0
8	1	VIS	_	8	_	_	_	0
9	1	TRT*VIS	1	5	_	_	_	0.3333
10	1	TRT*VIS	1	6	_	_	_	0
11	1	TRT*VIS	1	8	_	_	_	0
12	1	TRT*VIS	2	5	_	_	_	0.3333
13	1	TRT*VIS	2	6	_	_	_	0
14	1	TRT*VIS	2	8	_	_	_	0
15	1	TRT*VIS	3	5	_	_	_	0.3333
16	1	TRT*VIS	3	6	_	_	_	0
17	1	TRT*VIS	3	8	_	_	_	0
18	1	TRT*VIS	4	5	_	_	_	-1
19	1	TRT*VIS	4	6	_	_	_	0
20	1	TRT*VIS	4	8	_	_	_	0
21	1	TANSGR1N	_	_	1	_	_	0
22	1	TANSGR1N	_	_	2	_	_	0
23	1	RACEGR1N	_	_	_	1	_	0
24	1	RACEGR1N	_	_	_	2	_	0
25	1	WGTGRBLN	_	_	_	_	1	0
26	1	WGTGRBLN	_	_	_	_	2	0
27	1	BASE	_	_	_	_	_	0

Interpretation of results

LSMESTIMATE

	STMTNO	EFFECT	LABEL	ESTIMATE	STDERR	DF	TVALUE	PROBT	ALPHA	LOWER	UPPER
1	1	TRT*VIS	TRT1+TRT2+TRT3 vs TRT4 at VIS1	-0.5579	1.7794	84	-0.31	0.7547	0.05	-4.0964	2.9806

ESTIMATE

	LABEL	ESTIMATE	STDERR	DF	TVALUE	PROBT	ALPHA	LOWER	UPPER
1	TRT1+TRT2+TRT3 vs TRT4 at VIS1	-0.5579	1.7794	84	-0.31	0.7547	0.05	-4.0964	2.9806

CONTRAST

	LABEL	NUMDF	DENDF	FVALUE	PROBF
1	TRT1+TRT2+TRT3 vs TRT4 at VIS1	1	84	0.10	0.7547

Any Questions?